Warm-up: Lagrange interpoláció és lineáris algebra

Lineáris algebra.
Mutassuk meg a példánkon (X={1,2,3,4,5}, Y={0,6,24,60,120}), hogy a Lagrange interpoláció kezelhető a határozatlan együtthatók módszerével.

In[1]:=
```
p = α0 + α1 x + α2 x^2 + α3 x^3 + α4 x^4;
```

Az α_j együtthatókat az interpolációs krit alapján számoljuk ki.

In[2]:=
```
X = {1, 2, 3, 4, 5}; Y = {0, 6, 24, 60, 120};
```

In[3]:=
```
es = Table[(p /. x → X[[j]]) == Y[[j]], {j, 1, 5}]
```

Out[3]=
```
{α0 + α1 + α2 + α3 + α4 == 0, α0 + 2 α1 + 4 α2 + 8 α3 + 16 α4 == 6, α0 + 3 α1 + 9 α2 + 27 α3 + 81 α4 == 24,
 α0 + 4 α1 + 16 α2 + 64 α3 + 256 α4 == 60, α0 + 5 α1 + 25 α2 + 125 α3 + 625 α4 == 120}
```

In[4]:=
```
Solve[es]
```

Out[4]=
```
{{α0 → 0, α1 → -1, α2 → 0, α3 → 1, α4 → 0}}
```

In[5]:=
```
InterpolatingPolynomial[Transpose[{X, Y}], x] // Expand
```

Out[5]= $-x + x^3$

Newton előállítása osztott differenciákkal (rekurzív)

A Lagrange alaknál új adat hozzávételére nincs lehetôség ⟶ Newton féle elôállítás osztott differenciák segítségével

In[6]:=
```
Clear[OD, x0, y0, x1, y1]
```

In[7]:=
$$\text{NewtInterp}[\text{x\_List}, \text{y\_List}, \text{var\_}] := \sum_{j=1}^{\text{Length}[x]} \text{OD}[\text{Take}[x, j], \text{Take}[y, j]] \prod_{i=1}^{j-1} (\text{var} - x[[i]])$$

In[8]:=
```
NewtInterp[{x0, x1, x2}, {y0, y1, y2}, x]
```

Out[8]=
```
OD[{x0}, {y0}] + (x - x0) OD[{x0, x1}, {y0, y1}] + (x - x0) (x - x1) OD[{x0, x1, x2}, {y0, y1, y2}]
```

In[9]:=
```
OD[{x_}, {y_}] := y;
OD[x_List, y_List] :=
 (OD[Drop[x, 1], Drop[y, 1]] - OD[Drop[x, -1], Drop[y, -1]]) / (Last[x] - First[x])
```

In[11]:=
```
OD[{x0, x1}, {y0, y1}]
```

Out[11]=
$$\frac{-y0 + y1}{-x0 + x1}$$

In[12]:=
```
NewtInterp[{x0, x1, x2}, {y0, y1, y2}, x]
```

Out[12]=
$$y0 + \frac{(x - x0)\,(-y0 + y1)}{-x0 + x1} + \frac{(x - x0)\,(x - x1)\left(-\frac{-y0 + y1}{-x0 + x1} + \frac{-y1 + y2}{-x1 + x2}\right)}{-x0 + x2}$$

Konkrét Lagrange polinom előállításhoz:

In[13]:=
```
NewtInterp[{1, 2, 3, 4, 5}, {0, 6, 24, 60, 120}, x]
```

Out[13]= $6\,(-1+x)+6\,(-2+x)\,(-1+x)+(-3+x)\,(-2+x)\,(-1+x)$

```
Expand[%]
```

Megjegyzés. Séma az f[x0], f[x0;x1], f[x0;x1;x2], f[x0;x1;x2;x3],f[x0;x1;x2;x3;x4] osztott differencia sorozat kiszámításához.

$$L_4[x] = ((((0)+6\,(-1+x))+6\,(-2+x)\,(-1+x))+1\,(-3+x)\,(-2+x)\,(-1+x))+0\,(-1+x)\,(-2+x)\,(-3+x)\,(-4+x)\,(-5+x)$$

x_i	y_i	1 .	2.	3.	4.
1	0				
		6			
2	6		6		
		18		1	
3	24		9		0
		36		1	
4	60		12		
		60			
5	120				

In[14]:=
```
InterpolatingPolynomial[Transpose[{{1, 2, 3, 4, 5}, {0, 6, 24, 60, 120}}], x]
```

Out[14]= $(-1+x)\,(6+(-2+x)\,(3+x))$

In[15]:=
```
InterpolatingPolynomial[Transpose[{{1, 2, 3}, {2, 5, 10}}], x]
```

Out[15]= $2+(-1+x)\,(1+x)$

A Mathematica InterpolatingPolynomial függvény is így dolgozik.

In[16]:=
```
InterpolatingPolynomial[Transpose[{{x0, x1, x2}, {y0, y1, y2}}], x]
```

Out[16]= $$y0+(x-x0)\left(\frac{-y0+y1}{-x0+x1}+\frac{(x-x1)\left(-\frac{-y0+y1}{-x0+x1}+\frac{-y1+y2}{-x1+x2}\right)}{-x0+x2}\right)$$

Illusztráljuk a példánkon (X={1,2,3,4,5}, Y={0,6,24,60,120}), hogy az osztott differenciák tkp. egy alkalmas másik bázisban vett koordinátasorral egyeznek meg.
Megoldás. Legf. 4-edfokú polinomok lin tere P_4. Az intterpolációs feltételrendszer ekvivalens egy Vandermonde együtthatómátrixú lineáris egyenletrendszerrel.

In[17]:=
```
V[l_] := Table[l[[i]]^j, {i, Length[l]}, {j, 0, Length[l] - 1}]
```

In[18]:=
```
V[{1, 2, 3, 4, 5}] // MatrixForm
```

Out[18]//MatrixForm=
$$\begin{pmatrix} 1 & 1 & 1 & 1 & 1 \\ 1 & 2 & 4 & 8 & 16 \\ 1 & 3 & 9 & 27 & 81 \\ 1 & 4 & 16 & 64 & 256 \\ 1 & 5 & 25 & 125 & 625 \end{pmatrix}$$

```
In[19]:= U = {u0, u1, u2, u3, u4};

In[20]:= V[{1, 2, 3, 4, 5}][[2]].U

Out[20]= u0 + 2 u1 + 4 u2 + 8 u3 + 16 u4

In[21]:= Inverse[V[{1, 2, 3, 4, 5}]].{0, 6, 24, 60, 120}

Out[21]= {0, -1, 0, 1, 0}

In[22]:= LinearSolve[V[{1, 2, 3, 4, 5}], {0, 6, 24, 60, 120}]

Out[22]= {0, -1, 0, 1, 0}

In[23]:= %.Table[x^(j-1), {j, 5}]
```

Out[23]= $-x + x^3$

A másik bázis elemei, bázisátmenetmátrix (BTM)

```
In[24]:= B = {1, (x - 1), (x - 1) (x - 2), (x - 1) (x - 2) (x - 3), (x - 1) (x - 2) (x - 3) (x - 4)}

Out[24]= {1, -1 + x, (-2 + x) (-1 + x), (-3 + x) (-2 + x) (-1 + x), (-4 + x) (-3 + x) (-2 + x) (-1 + x)}

In[25]:= (BTM = Transpose[Table[PadRight[CoefficientList[Expand[B[[j]]], x], 5], {j, 5}]]) // TableForm
```

Out[25]//TableForm=

1	-1	2	-6	24
0	1	-3	11	-50
0	0	1	-6	35
0	0	0	1	-10
0	0	0	0	1

```
In[26]:= Inverse[BTM].Inverse[V[{1, 2, 3, 4, 5}]].{0, 6, 24, 60, 120}

Out[26]= {0, 6, 6, 1, 0}
```

Tehát:

```
In[27]:= %.B

Out[27]= 6 (-1 + x) + 6 (-2 + x) (-1 + x) + (-3 + x) (-2 + x) (-1 + x)
```

Ekvidisztáns alappontrendszer, progresszív differenciák

```
In[29]:= Clear[x0, x1, y0, y1]

In[30]:=
OD[{x0, x0 + h}, {f0, f1}]
```

Out[30]= $\frac{-f0 + f1}{h}$

```
In[31]:= OD[{x0, x0 + h, x0 + 2 h}, {f0, f1, f2}] // Simplify
```

Out[31]= $\frac{f0 - 2\, f1 + f2}{2\, h^2}$

```
In[32]:= OD[{x0, x0 + h, x0 + 2 h, x0 + 3 h}, {f0, f1, f2, f3}] // Simplify
```

Out[32]= $\frac{-f0 + 3\, f1 - 3\, f2 + f3}{6\, h^3}$

Progresszív differenciák

```
In[33]:= Clear[PD];
```

In[34]:=
```
PD[{y0_}] := y0;
PD[y_List] := PD[Drop[y, 1] - Drop[y, -1]]
```

In[36]:=
```
F0 = {f0, f1, f2, f3};
```

In[37]:=
```
Table[PD[Take[F0, j]], {j, Length[F0]}]
```

Out[37]=
```
{f0, -f0 + f1, f0 - 2 f1 + f2, -f0 + 3 f1 - 3 f2 + f3}
```

Nem kell h^n -nel osztani. Indoklás:

Legyen a vizsgált pont x:=x0+th és nézzük pl. a kvadratikus tagot:

$(x - x0)\,(x - x1) = t\,(t - 1)\,h^2 \qquad \Longrightarrow \qquad$ f[x0;x1;x2](x-x0)(x-x1)

$(f0 - 2\,f1 + f2)\,/\,(2\,h^2)\;h^2\;t\,(t - 1) = PD[f0,\,f1,\,f2]\binom{t}{2}$

Csak az ált. bin eüh és a progreszív differencia kell.

Sin[π / 9] közelítése interpolációval. Lépésköx: π/ 12. x0=0

In[38]:=
```
F0 = N[Sin[{0, Pi / 12, Pi / 6, Pi / 4, Pi / 3}], 4];
```

In[39]:=
```
F0
```

Out[39]=
```
{0, 0.2588, 0.5000, 0.7071, 0.8660}
```

In[40]:=
```
Table[PD[Take[F0, j]], {j, 1, Length[F0]}]
```

Out[40]=
```
{0, 0.2588, -0.018, -0.016, 0.002}
```

Interpolációs formula progresszív differenciákkal és ált. binomiális együtthatókkal

In[41]:=
$$\text{Newt2Interp}[x0\_,\ t\_,\ DL\_List] := \sum_{j=0}^{\text{Length}[DL]-1} \text{Binomial}[t,\ j]\ DL[[j+1]]$$

In[42]:=
```
Newt2Interp[0, 4 / 3, Table[PD[Take[F0, j]], {j, 1, Length[F0]}]]
```

Out[42]=
```
0.3420
```

```
Sin[Pi / 9] // N
```

Out[43]=
```
0.34202
```

Binomiális együttható általánosítása

In[44]:=
```
Binomial[4 / 3, 3]
```

Out[44]= $-\frac{4}{81}$

In[45]:=
```
(4 / 3) (1 / 3) (-2 / 3) / 6
```

Out[45]= $-\frac{4}{81}$

A formula az eredeti általános interpolációs formulának valóban speciális esete

```
Clear[x0]
```

In[46]:=
```
NewtInterp[{x0, x0 + h, x0 + 2 h, x0 + 3 h, x0 + 4 h}, {f0, f1, f2, f3, f4}, x] /. {x → x0 + t h} /.
  {t → 4 / 3} // Simplify
```

Out[46]= $\frac{1}{243} (-10\ f0 + 160\ f1 + 120\ f2 - 32\ f3 + 5\ f4)$

In[47]:= $\sum_{j=0}^{4}$ `Binomial[4 / 3, j] PD[Take[{f0, f1, f2, f3, f4}, j + 1]] // Simplify`

Out[47]= $\frac{1}{243} (-10\ f0 + 160\ f1 + 120\ f2 - 32\ f3 + 5\ f4)$

---

*CRM

- **Base Algorithm: (Newton)**

Consider only the first two and solve it

r=r1'+σm1 (m1m2)
σ=(r2-r1') c (m2)
c=m1^(-1) (m2)

In[48]:= `X = {1, 2, 3}; Y = {2, 5, 10};`

In[49]:= `PolynomialExtendedGCD[(x - 1), x - 2, x]`

Out[49]= {1, {1, -1}}

In[50]:= `c = 1; r1 = 2; r2 = 5; σp = (r2 - (r1)) c`

Out[50]= 3

In[51]:= `σ = PolynomialRemainder[%, x - 2, x]`

Out[51]= 3

In[52]:= `r1 + σ (x - 1) // Expand`

Out[52]= $-1 + 3\ x$

In[53]:= `PolynomialExtendedGCD[(x - 1) (x - 2), x - 3, x]`

Out[53]= $\left\{1, \left\{\frac{1}{2}, -\frac{x}{2}\right\}\right\}$

In[54]:= $\left\{\frac{1}{2}, -\frac{x}{2}\right\}$`.{(x - 1) (x - 2), x - 3} // Expand`

Out[54]= 1

In[55]:= `c = 1 / 2; r1 = 3 x - 1; r2 = 5; σp = (10 - (r1)) c`

Out[55]= $\frac{1}{2} (11 - 3\ x)$

In[56]:= `σ = PolynomialRemainder[%, x - 3, x]`

Out[56]= 1

In[57]:= `r1 + σ (x - 1) (x - 2) // Expand`

Out[57]= $1 + x^2$